

Modellek

Az előbbieken ismertetett probléma megoldását nem egy lépésben oldottam meg. Először is definiáltam különböző feladatosztályokat, amiket fokozatosan bővítettem. A feladatosztályokhoz kapcsolódóan pedig MILP modelleket írtam fel. Ezeket mutatom be jelen fejezetben.

Feladatosztály 1

Ez az osztály foglalkozott az alapesetrel, amiben még nincsen a buszoknak depója, vagyis az a hely, ahonnan a járatok elvégzése előtt elindul és ahova a járatok elvégzése után visszatér. Ebben az esetben még onnan indult egy busz ahonnan akart.

A kiindulási helyzetet is több irányból közelítettem meg. Felírtam két modellt (Modell1A és Modell1B), amelyek közül a későbbiekben már csak az egyiket fejlesztettem tovább.

Bemeneti adatok

A főbb bemeneti adatok a buszok, a buszok járatai, valamint a járatok kezdő és végpontjával szolgáló helyek. Ezeket az adatokat halmazokba rendeztem, és különböző kiegészítő paramétereket definiáltam hozzájuk.

A **Helyek** halmaz azon földrajzi helyeket tartalmazza, amelyek a buszjáratok kezdő és végpontját adják. A halmaz elemeinek egymáshoz viszonyított értékeit a **tav** és az **ido** paraméterek írják le. A **tav** néven definiált paraméter az adott helyiségek közötti kilométerben megadott távolságot mutatja meg. Az **ido** nevű paraméter a két helyiség közötti átlagos percben számolt átjutási időt jelzi.

Az elvégzendő buszjáratok a **Jaratok** halmazban találhatóak. A halmazhoz tartozó paraméterek a következők:

- **honnan** - a járat kiindulási helye a **Helyek** halmazból
- **hova** - a járat befejezésének helye a **Helyek** halmazból
- **mikortol** - a járat percben megadott kezdési időpontja
- **meddig** - a járat percben megadott befejezésének időpontja

- **jaratszám** - a végrehajtandó buszjáratok száma.

A rendelkezésre álló járműveket a **Buszok** halmaz fogja egybe. Az utazáshoz **buszszám** mennyiségű busz áll rendelkezésre.

Modell 1A

Változók

A modell két bináris változót használ. Ezek közül az egyik a `hozzarendel{Jaratok, Buszok}` változó, melynek értéke jelzi azt, hogy melyik buszjáratot ténylegesen melyik busz végzi el. Amennyiben az adott járat az adott buszhoz tartozik, akkor a változó értéke 1, ellenkező esetben 0.

Két buszjárat közötti átjárást az `atmenet{Buszok, Jaratok, Jaratok}` változó értékével fejeztem ki. Ha egy b busz végrehajt egy j_1 járatot és utána közvetlenül a j_2 járatot is, akkor az `atmenet[b, j1, j2]` változó értéke 1 lesz a későbbi ehhez kapcsolódó korlátozások hatására, minden más esetben pedig a 0 értéket veszi fel.

Bevezettem továbbá egy halmazt azoknak a járatpároknak, amiket nem lehet egy buszhoz rendelni, mivel ütközés lépne fel a végrehajtás és köztes utazások összes időtartamában. Ezek a párok a `Kulonbozobusz` halmaz elemeiként jelennek meg a modellben.

```
set Kulonbozobusz := setof{j in Jaratok, j2 in Jaratok:
    mikortol[j]<=mikortol[j2] &&
    mikortol[j2]<meddig[j]+ido[hova[j], honnan[j2]]
    && j!=j2} (j, j2);
```

Ha például veszünk két egymástól eltérő j_1 és j_2 járatot, és j_2 járat kezdési időpontja előbb van, mint ahogy egy busz az előző j_1 járata után odaérne a j_2 járat kiindulási helyére, akkor ez azt jelenti, hogy a két járatot külön busszal lehet csak végrehajtani. Ilyen párokból áll az előbb említett halmaz.

Korlátozások

A `JaratokElvegze` korlátozás minden járathoz hozzárendel egy buszt, így biztosítja, hogy minden járat végre legyen hajtva.

```
s.t. JaratokElvegze {j in Jaratok}: sum {b in Buszok} hozzarendel[j, b]=1;
```

Csak különböző buszokkal elvégezhető járatpárokat nem rendelhetünk ugyanahhoz a buszhoz. Ennek érdekében jött létre az *OsszeferhetetlenJaratok* nevű korlátozás. A *Kulonbozobusz* halmazból vett járatpárt nem lehet egyszerre az adott buszhoz rendelni, legfeljebb a pár egyik tagját.

```
s.t. OsszeferhetetlenJaratok{(j,j2) in Kulonbozobusz, b in Buszok}:  
    hozzarendel[j,b]+hozzarendel[j2,b] <= 1;
```

Amennyiben két járat hozzá van rendelve egy buszhoz és nincs a két járat között más köztes járat, akkor ez azt jelenti, hogy a busz az egyik járatból a másikba közvetlenül átmegy. Ilyen esetben az *atmenet* változó értéke 1 lesz, egyébként pedig 0. Ezt a megállapítást tükrözi az *AtmenetKorlatozas*.

```
s.t. AtmenetKorlatozas  
    {b in Buszok, j in Jaratok, j2 in Jaratok:mikortol[j2]>meddig[j] }:  
    atmenet[b,j,j2]  
    + sum {jkoztes in Jaratok: mikortol[jkoztes] >= meddig[j] &&  
    meddig[jkoztes] <= mikortol[j2]} hozzarendel[jkoztes,b]  
    >= -1+hozzarendel[j,b]+hozzarendel[j2,b];
```

Az előbbieken bemutatott korlátozás két másikkal tovább élesíthető. (*AtmenetKorlatozas_elesito1*, *AtmenetKorlatozas_elesito2*).

Cél

A cél a buszjáratok közötti áthaladások távolságainak minimalizálása. A célfüggvény felírásakor összegezni kell a járatok közötti átmenetek kilométeradatait.

```
minimize Koztestav:  
    sum {b in Buszok, j1 in Jaratok, j2 in Jaratok}  
    tav[hova[j1],honnan[j2]]*atmenet[b,j1,j2];
```

Modell 1B

Változók

A változók felírásához szükség volt még néhány újdonságra Modell1A-hoz képest, ezért bevezetésre került egy *maxjarat* nevű paraméter. Ez az érték az egy nap egy busz által

elvégezhető járatok maximális számát jelöli. Ennek segítségével tudjuk definiálni a **Sorszam** elnevezésű halmazt, amely **maxjarat** mennyiségű elemet tartalmaz.

Ez a modell is két változóval dolgozik, azonban az előzőhöz képest itt már nem csak bináris változókról van szó. A **hozzarendel{Buszok, Sorszam, Jaratok}** nevű bináris változó 1 értéket akkor szolgáltat, amennyiben az adott busz az adott sorszámú járatként elvégzi az adott járatot. Minden más esetben a változó értéke 0 lesz.

A másik változó a **koztesutazas{Buszok, s in Sorszam: s!=maxjarat}** amit a modell arra használ, hogy az egyes járatok közötti áthaladások kilométerben kifejezett köztes távolságait számolhassuk vele. Értelmem szerűen a változó csak nemnegatív értéket vehet fel.

Korlátozások

Az egyik korlátozás itt is az előírt járatok mindegyikének teljesítését hivatott szavatolni. A nevében is ugyanolyan **JaratokElvegze**se korlátozás azt mondja ki, hogy minden járatot hozzá kell rendelni egy busz valahányadik járatának.

```
s.t. JaratokElvegze {j in Jaratok} : sum {b in Buszok, s in Sorszam}
    hozzarendel[b,s,j]=1;
```

Az **EgyHelyreEgyet** nevű korlátozás szerint minden busz adott sorszámú járatának csak egyetlen járatot lehet választani.

```
s.t. EgyHelyreEgyet {b in Buszok, s in Sorszam} : sum {j in Jaratok}
    hozzarendel[b,s,j]<=1;
```

Ki kell küszöbölni azt az esetet, hogy ha nincs egy járatnak s-edik járata, akkor ne lehessen s+1-edik sem. Ezt teszi meg az **EgyasmellettiJaratok** korlátozás.

```
s.t. EgyasmellettiJaratok{b in Buszok, s in Sorszam: s!=maxjarat}:
    sum{j in Jaratok}hozzarendel[b,s,j] >=
    sum{j in Jaratok}hozzarendel[b,s+1,j];
```

Mivel ebben a modellben az előzőhöz képest nincsen Kulonbozobusz halmaz, ami a járatok közötti ütközések elkerülését hivatott segíteni, így felírtam egy korlátozást, ami ugyanezt a funkciót látja el. Az adott busznak elegendő idő kell ahhoz, hogy a j járat befejezési helyétől átérjen a j2 járat kezdési helyére. Csak akkor rendelhető hozzá az adott buszhoz mindkét (j, j2) járat, amennyiben teljesülnek az Idokorlat-ban megfogalmazottak.

```
s.t. Idokorlat{b in Buszok, s in Sorszam, j in Jaratok, j2 in Jaratok:
s!=maxjarat}:

    mikortol[j2] >= (meddig[j]+ido[hova[j],honnan[j2]]) *
    (-1+hozzarendel[b,s,j]+hozzarendel[b,s+1,j2]);
```

A járatok közötti köztes kilométerek számolására is létre kellett hozni egy korlátozást, ami a KoztesKmBeallitas1 nevet kapta. Ha j járat után j2 következik a busz menetrendjében és nem egyezik a j járat befejezési helye a j2 járat kezdési helyével, akkor hozzáadódik a koztesutazas változó értékéhez a megfelelő távolságadat.

```
s.t. KoztesKmBeallitas1{b in Buszok, s in Sorszam, j in Jaratok, j2 in
Jaratok:s!=maxjarat}:

    koztesutazas[b,s] >= tav[hova[j],honnan[j2]]*
    (-1+hozzarendel[b,s,j]+hozzarendel[b,s+1,j2]);
```

Azonban az előbb említett korlátozás túl nagy táblázatnyi korlátozást generál, ami nagyban lassítja a modell végrehajtását, ezért két új korlátozás is bevezetésre került. Ezek a korlátozásokból generált táblázat sorainak és oszlopainak egybevonásával segítik a modell gyorsítását (KoztesKmBeallitas2, KoztesKmBeallitas3).

Cél

A cél itt is az átmeneti kilométerek minimalizálása, amit ennek a modellnek a célfüggvényében a koztesutazas nevű változó segítségével fejezhetünk ki.

```
minimize Koztestav:

    sum{b in Buszok, s in Sorszam: s!=maxjarat}
        koztesutazas[b,s];
```

Összegzés

Mivel Modell 1B még a gyorsító korlátozások ellenére is sokkal lassabban működött (ezt majd a Tesztelés fejezetben jobban kifejtem), mint Modell 1A, így az először bemutatottat fejlesztettem tovább.

Feladatosztály 2

Ebben az esetben már minden egyes busz rendelkezik egy olyan hellyel ahonnan a járatok előtt elindul és ahova a járatok elvégzése után visszatér. Azonban még nem vettem figyelembe az olyan buszokra jellemző attribútumokat, mint például az üzemanyag, ami jelen esetben mivel elektromos buszjáratokról van szó, az elektromos áram. A későbbiekben ez a tulajdonság töltöttségként fog megjelenni. Jelen esetben még a járatok elvégzéséhez szükséges üzemanyagszintet feltételezünk.

Bemeneti adatok

Az első feladatosztály bemeneti adatait ebbe az osztályba is átemeltem, azonban itt már minden buszhoz hozzárendeltem egy `depo` nevű paramétert, ami egy `Helyek` halmazbeli helyet jelöl, annak jelzéseképp, hogy honnan kezd, és hol végzi a járatait egy busz.

Modell 2

Változók

Modell 1A-hoz képest kettővel több, azaz négy bináris változóval dolgoztam Modell 2-ben.

A két új bináris változó, amit bevezettem az `elsőjarat{Jaratok, Buszok}` és `utolsójarat{Jaratok, Buszok}` nevet viselik. Azért hoztam őket létre, hogy legyen olyan kapcsolóm, amivel ki tudom fejezni, azt hogy egy adott járat egy busznak az első illetve az utolsó járata-e, vagyis elsőként illetve utolsóként azt a járatot végzi-e el az adott busz. A depóból induló illetve ide érkező járatot is vehetjük első illetve utolsó járatnak.

Korlátozások

Az alapmodellhez (Modell 1A) hasonlóan itt is szükség van a `JaratokElvegze`, az `OsszeferhetetlenJaratok`, és az `AtmenetKorlatozas` nevű korlátozásokra. A

modell továbbá tartalmaz az első és utolsó járatokkal kapcsolatos és azokhoz tartozó redundánsan működő korlátozásokat is. A következőkben ezeket mutatom be.

A buszok első és utolsó járatainak meghatározásához hasonló korlátozások lettek létrehozva, így ezekről párhuzamosan tesztek említést. Akkor lehet egy járat első illetve utolsó járata egy busznak, ha azt a járatot az adott busz végzi el. Ellenkező esetben, ha a `hozzarendel[j,b]` változó értéke 0, akkor `ElsoHozzarendeles` és az `UtolsoHozzarendeles` korlátozásokban használt `elsojarat` és `utolsojarat` változók értéke is szükségképpen 0 lesz.

```
s.t. ElsoHozzarendeles{b in Buszok, j in Jaratok}:  
    elsojarat[j,b] <= hozzarendel[j,b];  
s.t. UtolsoHozzarendeles{b in Buszok, j in Jaratok}:  
    utolsojarat[j,b] <= hozzarendel[j,b];
```

Az előbbiekkal redundáns a következő két korlátozás, melyek szerint csak akkor lehet egy busznak első illetve utolsó járatáról beszélni, ha van bármilyen járat hozzárendelve ahhoz a buszhoz.

```
s.t. ElsoHozzarendeles2{b in Buszok}:  
    sum{j in Jaratok} elsojarat[j,b] <= sum{j in Jaratok} hozzarendel[j,b];  
s.t. UtolsoHozzarendeles2{b in Buszok}:  
    sum{j in Jaratok} utolsojarat[j,b] <= sum{j in Jaratok} hozzarendel[j,b];
```

A `LegfeljebbEgyElso` és `LegfeljebbEgyUtolso` nevű korlátozások szerint minden busznak maximum egy első és utolsó járata lehet.

```
s.t. LegfeljebbEgyElso{b in Buszok}:  
    sum{j in Jaratok} elsojarat[j,b] <= 1;  
s.t. LegfeljebbEgyUtolso{b in Buszok}:  
    sum{j in Jaratok} utolsojarat[j,b] <= 1;
```

Amennyiben legalább egy járat hozzá van rendelve egy buszhoz, akkor már mindenképpen szükséges a busz első illetve utolsó járatáról beszélni (SzuksegesElso, SzuksegesUtolso).

```
s.t. SzuksegesElso{b in Buszok}:  
    sum{j in Jaratok} elsojarat[j,b] >= sum{j in Jaratok} hozzarendel[j,b] /  
    card(Jaratok);  
s.t. SzuksegesUtolso{b in Buszok}:  
    sum{j in Jaratok} utolsojarat[j,b] >= sum{j in Jaratok} hozzarendel[j,b] /  
    card(Jaratok);
```

Egy olyan járat, ami a másiknál később kezdődik, és a buszhoz van rendelve, nem lehet egy busz első járata, valamint egy olyan járat, ami a másiknál korábban kezdődik, és a buszhoz van rendelve, nem lehet egy busz utolsó járata. Ezeket a feltételeket úgy illesztettem bele a modellbe, hogy két úgynevezett „nagy M” korlátozást hoztam létre *KesobbiNemElso* és *KorabbiNemUtolso* néven. Ezekhez a korlátozásokhoz szükséges volt felvenni egy kellően nagy értékű M paramétert.

```
s.t. KesobbiNemElso  
    {b in Buszok, j in Jaratok, j2 in Jaratok: mikortol[j]>mikortol[j2]}:  
        elsojarat[j,b] <= 0 + M * (1- hozzarendel[j2,b]);  
s.t. KorabbiNemUtolso  
    {b in Buszok, j in Jaratok, j2 in Jaratok: mikortol[j]<mikortol[j2]}:  
        utolsojarat[j,b] <= 0 + M * (1- hozzarendel[j2,b]);
```

Cél

A cél ugyanaz, mint a Modell 1A-ban megfogalmazott. A célfüggvényben összeszámoltam az összes járatok közötti távolságot, valamint azokat az utakat is, amíg a busz a depóból az első járata kezdési helyére ér, illetve az utolsó járata után a depóba visszaér.


```
minimize Koztestav:
sum {b in Buszok, j1 in Jaratok, j2 in Jaratok}
tav[hova[j1],honnan[j2]]*atmenet[b,j1,j2]
+
sum {b in Buszok, j in Jaratok} elsojarat[j,b]*tav[depo[b],honnan[j]]
+
sum {b in Buszok, j in Jaratok} utolsojarat[j,b]*tav[hova[j],depo[b]];
```

Feladatosztály 3

Az előzőekben még csak az általános buszokkal kapcsolatos alapokat írtam bele a modelljeimbe, azonban jelen osztálytól kezdődően már lépésenként elkezdtem az elektromos buszok tulajdonságait is figyelembe venni. Ilyen attribútum például a töltöttség. A buszok fogyasztásával illetve töltésével jelen esetben még nem foglalkoztam, azok már másik feladatosztályokhoz tartoznak.

Bemeneti adatok

A második feladatosztály továbbfejlesztéseként az összes ottani bemeneti adat itt is szerepel, viszont ehhez a modellhez hozzáadtam két újdonstágot is.

Ha egy járatról beszélünk, akkor, az nem feltétlen csak egy két pont közötti távolság lehet, hanem egy ennél hosszabb, kacsaringósabb út is. Ennek a számszerűsítésére hoztam létre a **Jaratok** halmazhoz tartozó **tav2** paramétert, ami egy járat két végpontja közötti valós hosszának kilométeradatát tartalmazza.

Mivel már elektromos buszokról beszélünk, így figyelembe kell venni azt is, hogy minden busznak van egy olyan határértéke, hogy hány kilométert tud menni egyetlen töltéssel. Ennek kapcsán szükség van erre az adatra is, ami ezt a buszokra vonatkozó számot jelzi. is Jelen esetben ez a **maxtoltes** nevű paraméter.

Modell 3

Változók

Többek között az előző modellbeli négy bináris változót használtam ebben a modellben is, de ezeken felül kellett még egy változó is, ami azt az értéket jelenti, hogy egy busz a járatok elvégzése közben hány kilométert tett meg. Ez a változó a modellben az `osszhasznalat{Buszok}` nevet kapta.

Korlátozások

Alapul szolgáltak a már Modell 1A és Modell 2-ben felépített korlátozások. Azok a járatok elvégzését, az összeférhetetlen járatok kiküszöbölését és a közvetlen átmenetek jelzését biztosították valamint meghatározták a buszok első és utolsó járatait.

Figyelembe kellett venni azonban azt is, hogy egy busz csak annyi kilométert tehet meg amennyit a töltöttsége enged, ezért szükség volt további két korlátozásra. Először is a megtett utakat úgy lehetett kiszámolni, hogy összeadtam a járatok valós hosszát, az átmeneti kilométereket és a depóból illetve a depóba jutás távolság adatait (`OsszhasznalatKiszamitas`), majd pedig felírtam azt a korlátozást, ami kimondja, hogy ez a szám nem lehet nagyobb, mint az egy töltéssel megtehető kilométerek száma (`MaxHasznalat`).

```
s.t. OsszhasznalatKiszamitas {b in Buszok}:  
    sum {j1 in Jaratok, j2 in Jaratok}  
    tav[hova[j1],honnan[j2]]*atmenet[b,j1,j2]  
    +  
    sum {j in Jaratok}elsojarat[j,b]*tav[depo[b],honnan[j]]  
    +  
    sum {j in Jaratok}utolsojarat[j,b]*tav[hova[j],depo[b]]  
    +  
    sum {j in Jaratok} hozzarendel[j,b]*tav2[j]  
    = osszhasznalat[b];
```

```
s.t. MaxHasznalat{b in Buszok}:  
    osszhasznalat[b] <= maxtoltes[b];
```

Cél

A modell célja továbbra is a köztes kilométerek minimalizálása.

Feladatosztály 4

A harmadik feladatosztályban bevezetésre került töltöttség kiegészítéseként most már a buszok energiafelhasználásával is foglalkoztam, de ebben az esetben még nem vettem figyelembe a töltések szükségességét.

Bemeneti adatok

Az előzőekben felhasznált bemeneti adatokat ez az osztály is átvette, azonban az ott megismert `maxtoltes` paraméter mértékegysége a továbbiakban, nem kilométerben, hanem Watt-ban megadott értéként fog szerepelni a modellekben. Ezt az osztályt a buszoknak az energiahasználatával bővítettem tovább, ezért az adatok között egy ezt leíró paraméterre is szükség volt. Ez a paraméter a `fogyasztas`, ami a buszoknak a kilométerenkénti Watt-ban mért energiafogyasztását mutatja meg.

Modell 4

Változók

Modell 3 továbbfejlesztéseként az ottani változókra ebben a modellben is szükségem volt. Újdonságként szerepel az `osszfogyasztas{Buszok}` változó, ami csak nemnegatív értéket vehet fel, mivel annak az összes watt mennyiségnek a tárolására szolgál, hogy az adott busz a járatai elvégzése során összesen mennyi energiát használt fel.

Korlátozások

A már megismert korlátozásokból kivettem a `MaxHasznalat` nevűt, és helyette felírtam két energiafogyasztáshoz kapcsolódó korlátozást. A `FogyasztasKiszamitas` segítségével meghatároztam a buszok összes energiafelhasználását. Ez az érték az `osszhasznalat` és a `fogyasztas` szorzatából jött ki.

```
s.t. FogyasztasKiszamitas{b in Buszok}:  
    osszfogyasztas[b] = osszhasznalat[b]*fogyasztas[b];
```

A **MaxFogyasztas** nevű korlátozás szerint a busz nem fogyaszthat több energiát a járatok elvégzése során, mint amennyit egy töltéssel kap.

```
s.t. MaxFogyasztas{b in Buszok}:  
    osszfogyasztas[b] <= maxtoltes[b];
```

Cél

Ettől a modelltől kezdődően már nem a köztes távolságok kilométereinek minimalizálása a cél, hanem a buszok összes energiafelhasználásának minimalizálása.

```
minimize Buszok_osszes_fogyasztasa: sum {b in Buszok} osszfogyasztas[b];
```